Paper 4937-2020

Mixed-Effects Models and Complex Survey Data with the GLIMMIX Procedure

L. Grisell Diaz-Ramirez, University of California, San Francisco;
Bocheng Jing, Northern California Institute for Research and Education;
Kenneth E. Covinsky, University of California, San Francisco;
W. John Boscardin, University of California, San Francisco

ABSTRACT

Longitudinal data with repeated measures are of great interest in clinical and epidemiology research and are often analyzed using a mixed-effects model. These data are frequently collected through complex survey methods. Thus, the survey features probability weights, clusters, and strata should be included in the analysis to obtain accurate population-based estimates. Fitting mixed models with survey data is still an area of active research, and currently there is no survey analysis procedure in SAS® for mixed models. As a workaround, researchers can use some of the options available in the SAS GLIMMIX procedure. However, when there are many strata or clusters, the model tends to be computationally intensive and complicated to optimize. In this paper, we explore different methods to account for survey features in mixed models using PROC GLIMMIX. We compare the parameter estimates, 95% confidence intervals, and running time among models, and make some recommendations on the specification of the final model. We illustrate the proposed methods with the Health and Retirement Study survey data. The data analysis for this paper was generated using SAS/STAT® 15.1 for Microsoft Windows.

INTRODUCTION

Longitudinal studies that gather data from the same subjects repeatedly over time are widely used in the clinical and epidemiology fields. These studies allow health researchers to learn about individual patterns occurring over long periods. Data from large-scale nationally representative survey samples are particularly useful for longitudinal studies, as respondents are usually followed up over extended period of times, a broad range of health and demographic information can be collected, and a large number of respondents can be included.

Mixed-effects models are often used in analyzing longitudinal data with repeated measures. These models allow the specification of both fixed and random effects. The fixed effects describe how the population means differ across subject characteristics, whereas the random effects capture the variability among subjects or other units. In longitudinal survey data, where we have multiple observations per subject across waves and subjects are grouped into clusters and strata, the random effects allows us to account for the correlation of observations within the same subject and the correlation of subjects within the same cluster or strata.

When longitudinal data collected using complex survey methods are analyzed with mixed-effects models, you need to account for the survey features probability weights, clusters, and strata. Failure to account for survey features can lead to biased population estimates and making wrong inferences about the population of study. However, fitting mixed models with survey data is still an area of active research, and currently there is no survey analysis procedure in SAS for mixed models.

Zhu (2014) showed how to use the GLIMMIX procedure to fit a weighted mixed model while accounting for clusters. The author used the quadrature algorithm to obtain the weighted likelihood and the empirical (sandwich) estimator to obtain the standard errors (SEs) for inferences on the fixed effects. However, as noted in this paper, mixed models with many levels of strata and clusters can be computationally intensive and complicated to optimize.

In this paper, we are building on Zhu's (2014) work by comparing how different methods of accounting for survey features affect the parameter estimates, 95% confidence intervals (CI), and processing time of the fitted mixed-effects model. The goal is to find a balance between obtaining accurate estimates with correct SEs and manageable processing time. We hypothesize that we could use the suboption FASQUAD, the PARMS statement, and the normalized weights to shorten the running time. We increase the complexity of the model sequentially by fitting: (1) a model with no weights, (2) a model with probability weights, (3) a two-step model that first uses probability weights and the FASQUAD suboption to obtain the fixed effects estimates and second uses normalized weights to correct the SEs of the fixed effects, (4) a model with probability weights and sandwich estimator—namely, the "Gold standard" model—, and (5) a two-step model that first uses FASQUAD suboption to obtain approximate covariance parameters estimates and second uses the PARMS statement with covariance estimates obtained with FASTQUAD. Based on these results, we make some recommendations on the specification of the final model. We illustrate the proposed methods with the extensively used Health and Retirement Study (HRS) survey data.

FITTING A MIXED-EFFECTS MODEL WITH PROC GLIMMIX AND SURVEY FEATURES

The following code shows how to fit a linear mixed-effects model with 2 splines, random intercepts and slopes, and the survey features probability weights and clusters (Zhu, 2014). We call this method the "Gold standard":

```
proc glimmix data=groups method=quadrature(qpoints=1) empirical=classical
noclprint;
   class procgroupmed (ref='0.PCI') newid clustervar;
   effect t5sas = spline(timeyrs5 / degree=1 knotmethod=list(5) basis=TPF);
   model memimp = procgroupmed intercept5 t5sas t5sas*procgroupmed
/solution cl ddfm=residual;
   random intercept / subject=clustervar type=chol;
   random intercept timeyrs5 / subject=newid(clustervar) weight=pre_wgtr
type=vc;
run;
```

In the code above:

- METHOD=QUADRATURE: fits a mixed-effects model using the Gauss-Hermite quadrature algorithm to obtain the weighted likelihood.
- QPOINTS=1: uses one quadrature point to obtain the likelihood function. Reducing the number of quadrature points reduces the amount of computation. This is not required, and you can let SAS decide the number to use by not specifying this option.
- EMPIRICAL=CLASSICAL: uses the sandwich estimator to obtain SEs for inferences on the fixed effects estimated by the weighted likelihood.
- NCLPRINT: suppresses the "Class Level Information" table and can help reduce memory and execution time.
- CLASS statement: specifies the categorical variables to be used in the model. The REF= option specifies the reference level for the categorical variable 'procgroupmed'.
- EFFECT statement: creates two linear splines with knot at 5.
- MODEL statement: fits a mixed model using categorical predictor 'procgroupmed' and splines created in the EFFECT statement. The DDFM=RESIDUAL option uses the residual

degrees of freedom for computing the denominator degrees of freedom. This method can help reduce the amount of memory needed to process the model.

- First RANDOM statement: incorporates random intercepts for the survey cluster variable ‘clustervar’. This accounts for the correlation of respondents within ‘clustervar’.
- TYPE=CHOL in the first RANDOM statement: specifies an unstructured variance-covariance matrix, parameterized through its Cholesky root, for ‘clustervar’ random effect. This was done to ensure numerical and statistical stability.
- Second RANDOM statement: incorporates random intercepts and random slopes for each respondent ‘newid’ nested within each ‘clustervar’. This accounts for the correlation among repeated observations within ‘newid’.
- TYPE=VC in second RANDOM statement: is the default covariance structure and specifies standard variance components. With this option, a distinct variance component is assigned to each effect.
- WEIGHT option in the second RANDOM statement: incorporates probability weights at the respondent level.

The variables in the code are:

- procgroupmed: categorical predictor with two levels that indicates type of surgical procedure performed.
- newid: respondent ID.
- clustervar: derived survey cluster variable with 104 levels.
- timeyrs5: time in years between procedure date and interview.
- t5sas: name for each linear spline, e.g. t5sas_1 and t5sas_2.
- memimp: composite memory score outcome.
- intercept5: indicator variable that takes a value of 0 for time before procedure and a value of 1 for time after procedure.
- pre_wgtr: survey probability weight for each respondent at the closest interview before procedure.

The code specified above shows the appropriate way of fitting a mixed-effects model while accounting for the survey features probability weights and clusters. However, as noted previously, when there are many strata or clusters, the model tends to be computationally intensive with long running times.

In the next section we explain alternative methods for fitting a mixed-effects model that account differently for the survey features probability weights and clusters.

DESCRIPTION OF ALTERNATIVE METHODS FOR FITTING A MIXED-EFFECTS MODEL WITH SURVEY FEATURES

We used five different methods to fit a linear mixed-effects model with two splines and random intercepts and slopes:

Method 1: this was a reference method that did not incorporate probability weights and the sandwich estimator. That is, method 1 did not use options: ‘METHOD=QUADRATURE’ and ‘EMPIRICAL=CLASSICAL’ in the PROC GLIMMIX statement, and it did not include weights in the second RANDOM statement as shown in the code below:

```
proc glimmix data= groups noclprint;
   class procgroupmed (ref='0.PCI') newid clustervar;
   effect t5sas = spline(timeyrs5 / degree=1 knotmethod=list (5)
basis=TPF);
   model memimp = procgroupmed intercept5 t5sas t5sas*procgroupmed
/solution cl ddfm=residual;
   random intercept / subject=clustervar type=chol;
   random intercept timeyrs5 / subject=newid(clustervar) type=vc
run;
```

Method 2: it incorporated probability weights at the respondent level, but it did not use the sandwich estimator. As the "Gold standard" method, method 2 used 'METHOD=QUADRATURE', and, in addition, it included the FASQUAD suboption as shown in the code below:

```
proc glimmix data= groups method=quadrature(fastquad qpoints=1)
noclprint;
   class procgroupmed (ref='0.PCI') newid clustervar;
   effect t5sas = spline(timeyrs5 / degree=1 knotmethod=list (5)
basis=TPF);
   model memimp = procgroupmed intercept5 t5sas t5sas*procgroupmed
/solution cl ddfm=residual;
   random intercept / subject=clustervar type=chol;
   random intercept timeyrs5 / subject=newid(clustervar)
weight=pre_wgtr type=vc;
run;
```

The FASTQUAD suboption in the 'METHOD=QUADRATURE' option invokes Pinheiro and Chao (2006) multilevel adaptive Gaussian quadrature algorithm. This algorithm reduces the number of random effects to three: one for the 'clustervar' random intercept and two for the 'newid' random slope and random intercept. Without this suboption, the number of random effects within each 'clustervar' is one for 'clustervar' random intercept plus two (random slope and random intercept) times the number of 'newid' nested within the 'clustervar'. For example, if there are 73 'newid' in one 'clustervar' the number of random effects for that clustervar equals: 1+2*73=147 random effects just for one 'clustervar'. In our study sample the number of 'newid' nested within 'clustervar' ranges from 1 to 73. As the number of random effect increases, the number of conditional log likelihoods computed also increases, which drives up the computational requirement. On the other hand, with FASQUAD the number of random effects does not increase with the number of respondents nested within 'clustervar'. Thus, the use of FASTQUAD reduces both computational demand and memory usage. However, FASTQUAD suboption is not currently available with the sandwich estimator which is needed to obtain the correct SEs.

Method 3: it incorporated normalized weights at the respondent level, and it did not use the sandwich estimator. The normalized weights were computed by dividing the probability weight of each respondent by its mean. This method was used by Jing et al. (2019) as a way to control the overestimation of the sample size when using only probability weights, which could lead to making wrong inferences on the estimates. Method 3 involved fitting two models. The first fitted model used the FASQUAD suboption with probability weights as in Method 2 to obtain the fixed effects estimates more quickly, and the second fitted model used the normalized weights (with FASQUAD) instead of the probability weights to obtain corrected SEs. Then, the 95% confidence intervals (CIs) of the fixed effects were computed as: Estimate ±1.96*SEs.

Method 4: this was the "Gold standard" model that incorporated probability weights and the sandwich estimator as shown in the previous section.

Model 5: this was a variation of Method 4 and involved fitting two models. The first fitted model used the FASQUAD suboption as in Method 2 to obtain the covariance parameters estimates more quickly, and the second fitted model used the PARMS statement with covariance estimates obtained with FASTQUAD. The code below shows how to fit the model with the PARMS statement:

```
proc glimmix data=groups method=quadrature(qpoints=1)
empirical=classical noclprint;
   class procgroupmed (ref='0.PCI') newid clustervar;
   effect t5sas = spline(timeyrs5 / degree=1 knotmethod=list(5)
basis=TPF);
   model memimp = procgroupmed intercept5 t5sas t5sas*procgroupmed
/solution cl ddfm=residual;
   random intercept / subject=clustervar type=chol;
   random intercept timeyrs5 / subject=newid(clustervar)
weight=pre_wgtr type=vc;
   parms (0.08903) (0.2721) (0.009698) (0.04074) / hold=1,2,3,4;
/* Covariance parameters obtained previously using FASQUAD option */
run;
```

Even though the model fitted with FASQUAD suboption might not have the right SEs for all the fixed effects, the covariance parameters and fixed effects estimates obtained are very similar to the ones obtained in the “Gold standard” method. So, we used the PARMS statement with the estimated covariance parameters to reduce the number of iterations and consequently the running time. In our model, we have four covariance parameters corresponding to: ‘clustervar’ random intercept, ‘newid’ nested within ‘clustervar’ random intercept, ‘newid’ nested within ‘clustervar’ random slope, and residual. The option HOLD in the PARMS statement specifies which parameter values PROC GLIMMIX should constrain to the specified values. In our example, all four covariance parameters are constrained.

EXAMPLE: HEALTH AND RETIREMENT STUDY DATA

We created a nationally representative cohort of 1,680 community-dwelling seniors enrolled in the Health and Retirement Study (HRS) who underwent coronary artery bypass graft (CABG) or percutaneous coronary intervention (PCI, or "stenting") at the age of 65 or older. The HRS is an ongoing longitudinal survey of a representative sample of all persons in the United States over age 50 that examines changes in health and wealth (Sonega et al., 2014). Respondents are interviewed every two years by phone and in-person interviews. It is sponsored by the National Institute on Aging (grant number NIA U01AG009740) and conducted by the University of Michigan. We used the public RAND HRS data file and the public HRS Imputation of Cognitive Functioning Measures.

The HRS sample incorporates special design features such as stratification (variable: strata), clustering (variable: cluster), and differential selection probabilities (variable: pre_wgtr). In this example, we have a total of 52 strata and 2 clusters. For our models, we created a derived variable ‘clustervar’ with 104 levels which was a combination of strata and clusters. That is, clustervar= strata + (100*cluster). We used this derived variable ‘clustervar’ instead of the original variables: strata and cluster to account for clustering. The probability weight ‘pre_wgtr’ was derived from the closest interview prior to the procedure.

The outcome was ‘Composite memory score’ (variable: memimp), a continuous variable derived following Wu et al. (2002) methodology at interviews before and after the procedure. The follow-up time was up to 15 years and encompassed interviews from 5 years prior to the procedure to 10 years after the procedure. Thus, analyses needed to account for repeated measures of cognitive score over time and the survey design features. The main predictor was type of surgery: CABG or PCI. The total sample size was 8,487 observations corresponding to 1,680 respondents.

The goal was to investigate whether respondents experienced a change in the slope of their cognitive trajectory following CABG or PCI, and whether the change in the slope was different depending on the procedure performed.

We fitted multivariable linear mixed-effects models of Composite memory score on time to interview. We used two piecewise linear splines with discontinuity and knot at the time of the procedure to model the pre and post trajectories. We included fixed intercepts and slopes for the predictors in the model, random intercepts for each ‘clustervar’ to account for the correlation of respondents within ‘clustervar’, and random intercepts and slopes for each respondent to account for the correlation among repeated observations.

COMPARISON OF METHODS FOR FITTING A MIXED-EFFECTS MODEL WITH SURVEY DATA

Table 1 shows the estimated fixed effects, corresponding 95% CIs, covariance parameters estimates, and running time across the five methods.

Effects	Method 1: model with no weights	Method 2: model with probability weights (FASQUAD)	Method 3: model with normalized weights (FASQUAD)	Method 4: "Gold standard" model with probability weights and sandwich estimator	Method 5: model with probability weights, sandwich estimator, and PARMS statement
	Fixed Effect Estimate (95% CI)				
Intercept	0.9808 (0.9451; 1.0166)	0.9620 (0.9232; 1.0008)	0.9620 (0.9266; 0.9974)	0.9620 (0.9212; 1.0028)	0.9608 (0.9217; 1.0000)
1.CABG	0.0094 (-0.0418; 0.0606)	0.0291 (-0.0255; 0.0837)	0.0291 (-0.0217; 0.0799)	0.0291 (-0.0225; 0.0807)	0.0294 (-0.0217; 0.0805)
intercept5	0.0037 (-0.0190; 0.0264)	0.0029 (0.0026; 0.0032	0.0029 (-0.0194; 0.0252)	0.0029 (-0.0213; 0.0271)	0.0029 (-0.0213; 0.0271)
t5sas_1	-0.0494 (-0.0575; -0.0413)	-0.0464 (-0.0525; -0.0403)	-0.0464 (-0.0544; -0.0384)	-0.0465 (-0.0549; -0.0381)	-0.0434 (-0.0518; -0.0350)
t5sas_1*CABG	0.0040 (-0.0071; 0.0151)	-0.0001 (-0.0098; 0.0096)	-0.0001 (-0.0111; 0.0110)	-0.0001 (-0.0099; 0.0096)	-0.0009 (-0.0106; 0.0089)
t5sas_2	-0.0224 (-0.0317; -0.0132)	-0.0315 (-0.0317; -0.0314)	-0.0315 (-0.0408; -0.0223)	-0.0315 (-0.0424; -0.0207)	-0.0316 (-0.0424; -0.0207)
t5sas_2*CABG	-0.0060 (-0.0202; 0.0083)	-0.0047 (-0.0050; -0.0045)	-0.0047 (-0.0190; 0.0095)	-0.0047 (-0.0204; 0.0110)	-0.0047 (-0.0204; 0.0110)
	Covariance Parameter Estimates				
clustervar	0.0616	0.0890	0.0890	0.0891	0.0890
newid(clustervar) intercept	0.1303	0.2721	0.2721	0.2722	0.2721
newid(clustervar) slope	0.0024	0.0097	0.0097	0.0097	0.0097
Residual	0.0705	0.0407	0.0407	0.0407	0.0407
Running time					
	16 seconds	1 minute and 50 seconds	2 minutes and 33 seconds	42 hours and 36 minutes	17 hours and 32 minutes

Table 1. Comparison of Parameters Estimates, 95% CIs, and Running Times across Five Methods for Fitting a Mixed-effects Model with Survey Features

In general, we found that fixed effects estimates and corresponding 95% CIs of “Gold standard” method 4 and its variants methods 3 and 5 are more similar amongst themselves than those obtained in methods 1 and 2. This was expected since, as the “Gold standard” method, methods 3 and 5 did account for probability weights. Additionally, Method 3 used normalized weights to control overestimating the sample size and consequently getting too small SEs, and method 5 used the sandwich estimator. The difference among these methods was that method 3 accelerated the execution time by using the normalized weights with the suboption FASQUAD, while method 4 used the PARMS statement with estimated covariance parameters to reduce the number of iterations and consequently the running time. In contrast, methods 1 and 2 did not use the sandwich estimator, and Method 1 did not consider probability weights.

The fixed effects estimates of Methods 2 and 3 with FASTQUAD suboption are almost the same as those in the “Gold standard” method. Likewise, the covariance parameter estimates of methods 2, 3, and 5 with FASQUAD suboption are almost the same as those obtained in the “Gold standard” method. This reassured us that using FASQUAD is a valid alternative to obtain correct fixed effects and covariance parameters estimates.

The 95% CIs of Method 5 that used the PARMS statement with constrained covariance parameters were the most similar to the 95% CI of the “Gold standard” method, followed by Method 3. In general, Method 3 had narrower 95% CIs than “Gold standard” method and Method 5 since the first one used the normalized weight (instead of the sandwich estimator) as a way to approximate the SEs. That is, despite the normalized weights can be used as an approximation to correct the SEs, there still might be some overestimation of the sample size which can cause smaller standard errors.

Method 2 that did not use the sandwich estimator had too-narrow 95% CIs for the fixed effects: “intercept5”, “t5sas_2”, and “t5sas_2*CABG”, while the 95% CIs of the rest of the effects were more similar to the “Gold standard” method. This finding suggest that even when the FASQUAD suboption is a valid alternative to obtain correct fixed effects and covariance parameters estimates, the researcher should be cautious about making inferences on the fixed effects if the sandwich estimator is not used.

It is worth noting that we also looked at the fixed effects estimates obtained in the model fitted with the normalized weights and FASQUAD (results not shown). Even though these estimates were not very different compared with those obtained in the “Gold standard” method, we decided to use the fixed effects estimates from Method 2 (with probability weights instead of normalized weight) since they were much more similar to the “Gold standard” model estimates.

Both methods 3 and 5 have a significantly shorter running time than the “Gold standard” method (17 hours, 32 minutes and 2 minutes, 33 seconds vs. 42 hours, 36 minutes respectively). This reduction in running time together with the finding that both methods 3 and 5 had almost the same parameter estimates and similar 95% CIs as the “Gold standard” make both methods a feasible alternative to fit mixed-effects models with long running times.

Despite the longer running time of Method 5 compared to Method 3 (17 hours, 32 minutes vs. 2 minutes, 33 seconds respectively), the first one had the advantage of allowing the estimation of the SEs of the predictions directly from the model. That is, you can save the results of the model fitted with Method 5 using the STORE statement, and then use the PLM procedure to get the predicted values (based on the fixed effects only) and their corresponding SEs for a specific sample. On the other hand, with Method 3 we can approximate the SEs of the fixed effects by using the normalized weights, but computing the SEs of the predictions would be more cumbersome.

To answer our research question of whether respondents experienced a change in the slope of their cognitive trajectory following CABG or PI and whether the change in the slope was different depending on the procedure performed, we focused on the fixed effects: “t5sas_1*CABG” and “t5sas_2*CABG”. The fixed effect “t5sas_1*CABG” described the difference in the slope between CAGB and PCI prior to the procedure, and the fixed effect “t5sas_2*CABG” described the difference of the change in slope (from prior to after procedure) between CABG and PCI. That is:

t5sas_1*CABG='pre-slope-CABG' – 'pre-slope-PCI'

t5sas_2*CABG=('post-slope-CABG' – 'pre-slope-CABG') – ('post-slope-PCI' – 'pre-slope-PCI')

Figure 1. shows the fixed effects estimates 't5sas_1*CABG' and 't5sas_2*CABG' and their corresponding 95% CIs across the five methods.



Methods:

Method 1: model with no weights
Method 2: model with probability weights (FASQUAD)
Method 3: model with normalized weights (FASQUAD)
Method 4: "Gold standard" model with probability weights and sandwich estimator
Method 5: model with probability weights, sandwich estimator, and PARMS statement

Figure 1. Comparison of Parameters Estimates of Change in Slopes between CABG and PCI and Corresponding 95% CIs across Five Methods for Fitting a Mixed-effects Model with Survey Features

All of the methods showed no significant difference in the slope prior to the procedure between CAGB and PCI (effect: 't5sas_1*CABG'). Despite the effect was not significant, its magnitude as well as its direction were more similar among methods 2 through 5. Method 1 had a larger and positive effect, whereas the other methods had a negative effect. It will be up to the clinical researcher to determine whether this difference in magnitude and direction is clinically relevant.

Similarly, regardless of the method used, and excluding Method 2, there was no significant difference in the change of the slope depending on the procedure performed (effect: 't5sas_2*CABG'). Method 2 had an unusually too-narrow 95% CIs (indistinguishable in the figure), because even though we incorporated the sampling weights, we did not use the sandwich estimator. This showed that caution should be used on making inferences on the fixed effects when the sandwich estimator is not used. For this effect, the size and direction among methods are more similar, but methods 2 through 5 have the greatest similarity.

In this study, the total sample size was 8,487 observations, and fitting the "Gold standard" model took 42 hours and 36 minutes. This running time might seem achievable. However, when the sample size increases, the increment in the computational time can be unmanageable. Thus, alternative methods that produce accurate estimates with correct SEs and feasible processing time are of great interest, particularly, with large sample sizes.

Finally, in our models we did not explicitly account for stratification by including strata as fixed effects. We ran some models with strata as fixed effects and found some numerical instability in the estimation of the model (i.e. some covariance parameters were missing). In addition, these models had a significantly larger running time (more than 10 days), so we did not pursue further this modeling strategy. Despite these models not having a clean convergence, in general, the fixed effects estimates and their 95% CI were similar to those obtained in the "Gold standard" method. Thus, we assumed that for this data and model we could ignore the effects of stratification.

CONCLUSION

Fitting a mixed-effects model that accounts for survey features is important to obtain accurate and precise estimates that can be used to generalize the results to the population of interest. However, models with many levels of strata and clusters can be computationally intensive, complicated to optimize, and consequently have long running times. The FASQUAD suboption, PARMS statement, and the use of normalized weights can be implemented as alternatives to produce reliable estimates while at the same time reduce the running time.

REFERENCES

1. Health and Retirement Study, (RAND HRS Longitudinal File 2016, V1) public use data set. Produced and distributed by the University of Michigan with funding from the National Institute on Aging (grant number NIA U01AG009740). Ann Arbor, MI, (2019).

2. Health and Retirement Study, (Imputation of Cognitive Functioning Measures: 1992 – 2016, Version 1.0) public use data set. Produced and distributed by the University of Michigan with funding from the National Institute on Aging (grant number NIA U01AG009740). Ann Arbor, MI, (October 2019).

3. Jing B, Diaz-Ramirez LG, Boscardin WJ, Lee AK, Lee SJ. Survival Estimates Prediction with Survey Data. Proceedings of the SAS® Global Forum 2019, Dallas, TX: SAS Institute Inc., Cary, NC. Available at: https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2019/3900-2019.pdf

4. Kiernan K, Tao J, Gibbs P. Tips and Strategies for Mixed Modeling with SAS/STAT® Procedures. Proceedings of the SAS® Global Forum 2012, Orlando, FL: SAS Institute Inc., Cary, NC. Available at http://support.sas.com/resources/papers/proceedings12/332-

2012.pdf

5. Pinheiro, JC and Chao EC. Efficient Laplacian and Adaptive Gaussian Quadrature Algorithms for Multilevel Generalized Linear Mixed Models. J Comput Graph Stat 2006; 15(1):58–81. https://doi.org/10.1198/106186006x96962.
6. RAND HRS Longitudinal File 2016, V1. Produced by the RAND Center for the Study of Aging, with funding from the National Institute on Aging and the Social Security Administration. Santa Monica, CA (May 2019).
7. SAS/STAT 15.1 User's Guide. The GLIMMIX Procedure. Example 49.19 Quadrature Method for Multilevel Model. Accessed February 23, 2020. Available at https://documentation.sas.com/?docsetId=statug&docsetTarget=statug_glimmix_examples24.htm&docsetVersion=15.1&locale=en
8. SAS/STAT 15.1 User's Guide. The GLIMMIX Procedure. Pseudo-likelihood Estimation for Weighted Multilevel Models. Accessed February 23, 2020. Available at https://go.documentation.sas.com/?docsetId=statug&docsetTarget=statug_glimmix_details25.htm&docsetVersion=15.1&locale=en
9. Sonnega A, Faul JD, Ofstedal MB, Langa KM, Phillips JW, Weir DR. Cohort profile: the Health and Retirement Study (HRS). Int J Epidemiol 2014; 43(2):576-585. https://doi.org/10.1093/ije/dyu067.
10. Wu Q, Tchetgen Tchetgen EJ, Osypuk TL, White K, Mujahid M, Glymour MM. Combining Direct and Proxy Assessments to Reduce Attrition Bias in a Longitudinal Study. Alzheimer Dis Assoc Disord 2013; 27(3):207-212. https://doi.org/10.1097/WAD.0b013e31826cfe90.
11. Zhu, M. Analyzing Multilevel Models with the GLIMMIX Procedure. Proceedings of the SAS® Global Forum 2014. Washington, DC: SAS Institute Inc., Cary, NC. Available at https://support.sas.com/resources/papers/proceedings14/SAS026-2014.pdf.

ACKNOWLEDGMENTS

This work was supported by the UCSF Pepper Center (P30 AG044281) and National Institute on Aging (R03AG059822), which promotes promising research aimed at better understanding and addressing late-life disability in vulnerable populations.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

L. Grisell Diaz-Ramirez
University of California, San Francisco
grisell.diaz-ramirez@ucsf.edu

Bocheng Jing
Northern California Institute for Research and Education (NCIRE)-The Veterans Health Research Institute
bocheng.jing@ncire.org

Kenneth E. Covinsky
University of California, San Francisco
ken.covinsky@ucsf.edu

W. John Boscardin
University of California, San Francisco
john.boscardin@ucsf.edu